



Example 5: Multivariate Multilevel Models for Non-Normal Outcomes in Mplus v. 8
(complete syntax and output for Mplus available electronically)

These (real) data come from a daily diary study that followed 41 male and female college students over a six-week period to examine within-person relationships among hyperarousal symptoms, alcohol use, and perpetration of intimate partner violence (IPV). To be eligible, potential participants had to be currently involved in a romantic relationship and have face-to-face contact with their partner at least once a week with no intentions of breaking up with their partner in the preceding six weeks. In addition, participants had to report using alcohol in the previous six weeks with no intention of abstaining from future alcohol use, and perpetrating or experiencing at least one instance of physical (e.g., pushing, shoving, slapping/punching, or choking), sexual (e.g., using threats or physical force to obtain sex), or psychological abuse (e.g., calling the partner stupid, worthless, or ugly) in the previous six months. Below are the distributions of the hyperarousal predictor (left) and intimate partner violence outcome (right) across all daily observations.



We will be examining several candidate models for IPV: Normal, Negative Binomial, Negative Binomial/Poisson Hurdle, and Two-Part Log. Although it may be somewhat suspect given its distribution above, for the sake of illustration we will treat hyperarousal as conditionally normal across models. We are, however, using the MLR estimator (i.e., robust maximum likelihood) that corrects the parameter standard errors (via the “sandwich” method) for non-normality.

Model 1: Normal Conditional Distribution

Later versions of the model generated this error:

```
*** ERROR in MODEL command
Unrestricted x-variables for analysis with TYPE=TWOLEVEL and
ALGORITHM=INTEGRATION must be specified as either a WITHIN or
BETWEEN variable. The following variable cannot exist on both
levels: HYPER
```

So we will use a latent variable work-around to trick Mplus: This is only necessary for non-normal distributions

```
TITLE:      Model 1: Normal Conditional Distribution
DATA:      FILE = Example5.csv;  ! Syntax in same folder as data
VARIABLE:
NAMES = PersonID hyper alc IPV;  ! List ALL variables in data file
USEVARIABLES = hyper IPV;        ! Variables included in this model
MISSING ARE ALL (-99);          ! Missing data identifier
CLUSTER = PersonID;             ! Level-2 ID
! No extra code here means each outcome is conditionally normal

ANALYSIS: TYPE = TWOLEVEL RANDOM;  ! In case of random slopes
ESTIMATOR = MLR;                  ! For non-normal hyper
```

```
MODEL:
%WITHIN%
hyper IPV;      ! L1 R: residual variances
IPV ON hyper;   ! L1 WP fixed effect
```

```
%BETWEEN%
! Move hyper random intercept to new latent variable
Fhyper BY hyper@1;
! Shut off hyper random intercept variance
hyper@0;
! IPV and hyper fixed intercepts (hyper=grand-mean-centered)
[hyper IPV];
! Shut off new latent variable intercept
[Fhyper@0];
! L2 G: random intercept variances
Fhyper IPV;
! L2 BP fixed effect using new latent variable
IPV ON Fhyper;
```

Variable	Intraclass Correlation	Variable	Intraclass Correlation	→ Only given for normal
HYPER	0.464	IPV	0.521	

Number of Free Parameters 8

Loglikelihood

H0 Value	-1385.261	
H0 Scaling Correction Factor	7.021	→ 1 means normal for MLR
H1 Value	-1385.261	
H1 Scaling Correction Factor	7.021	

	Estimate	S.E.	Est./S.E.	Two-Tailed P-Value
Within Level				
FOR EVERY UNIT INCREASE IN L1 WP HYPER, L1 WP IPV INCREASES BY .396: MORE HYPER THAN USUAL PREDICTS MORE IPV THAT DAY				
IPV ON				
HYPER	0.396	0.112	3.547	0.000
L1 RESIDUAL VARIANCE (NOTHING PREDICTING HYPER)				
Variances				
HYPER	0.132	0.024	5.577	0.000
L1 RESIDUAL LEFT OVER VARIANCE (IPV IS PREDICTED BY HYPER)				
Residual Variances				
IPV	0.783	0.242	3.230	0.001
Between Level				
INTERCEPT VARIANCE IN HYPER HAS BEEN MOVED TO FHYPER TO TRICK MPLUS				
FHYPER BY				
HYPER	1.000	0.000	999.000	999.000
FOR EVERY UNIT INCREASE IN L2 BP HYPER, L2 BP IPV INCREASES BY .850: MORE L2 MEAN HYPER THAN OTHERS ALMOST PREDICTS MORE L2 MEAN IPV THAN OTHERS				
IPV ON				
FHYPER	0.850	0.465	1.828	0.067
FIXED INTERCEPT FOR HYPER IS FIXED TO 0 TO CREATE GRAND-MC VERSION				
Means				
FHYPER	0.000	0.000	999.000	999.000
FIXED INTERCEPT FOR HYPER (HYPER IS UNCONDITIONAL SINCE IT IS NOT BEING PREDICTED BY ANYTHING)				
Intercepts				
HYPER	1.436	0.056	25.716	0.000
FIXED INTERCEPT FOR IPV (FOR SOMEONE WITH AVERAGE HYPER ACROSS DAYS)				
Intercepts				
IPV	0.561	0.146	3.848	0.000
L2 RANDOM INTERCEPT VARIANCE (NOTHING PREDICTING HYPER)				
Variances				
FHYPER	0.115	0.025	4.663	0.000
L2 RANDOM INTERCEPT VARIANCE LEFTOVER (IPV IS PREDICTED BY FHYPER)				
Residual Variances				
HYPER	0.000	0.000	999.000	999.000
IPV	0.793	0.460	1.725	0.084

Model 2: Negative Binomial Conditional Distribution

So we will use a latent variable work-around to trick Mplus:

```
TITLE:      Model 2: Negative Binomial (predict log of level-1 count)
DATA:      FILE = Example5.csv;  ! Syntax in same folder as data
VARIABLE:
NAMES = PersonID hyper alc IPV;  ! List ALL variables in data file
USEVARIABLES = hyper IPV;        ! Variables included in this model
MISSING ARE ALL (-99);           ! Missing data identifier
CLUSTER = PersonID;              ! Level-2 ID
COUNT = IPV (nb);              ! Now IPV is negative binomial (stretchy poisson)
```

! Do not need DEFINE for grand-mean-centering anymore (see below)

```
ANALYSIS: TYPE = TWOLEVEL RANDOM;  ! In case of random slopes
ESTIMATOR = MLR;                  ! For non-normal hyper
ALGORITHM = INTEGRATION;          ! Because of COUNT option
```

MODEL:

%WITHIN%

```
hyper;          ! L1 R: residual variance for hyper
IPV;            ! L1 dispersion for IPV
IPV ON hyper;   ! L1 WP fixed effect
```

%BETWEEN%

```
! Move hyper random intercept to new latent variable
Fhyper BY hyper@1;
! Shut off hyper random intercept variance
hyper@0;
! IPV and hyper fixed intercepts (hyper=grand-mean-centered)
[hyper IPV];
! Shut off new latent variable intercept
[Fhyper@0];
! L2 G: random intercept variances
Fhyper IPV;
! L2 BP fixed effect using new latent variable
IPV ON Fhyper;
```

Mplus is still confused, but lets us fit the model anyway...

*** WARNING in MODEL command

In the MODEL command, the following variable is a y-variable on the BETWEEN level and an x-variable on the WITHIN level. This variable will be treated as a y-variable on both levels: HYPER

	Estimate	S.E.	Est./S.E.	Two-Tailed P-Value
Within Level				
FOR EVERY UNIT HIGHER L1 WP HYPER, LOG OF L1 WP IPV IS HIGHER BY 1.348: MORE HYPER THAN USUAL PREDICTS MORE IPV THAT DAY (AS BEFORE)				
IPV ON				
HYPER	1.348	0.359	3.751	0.000
L1 RESIDUAL VARIANCE (NOTHING PREDICTING HYPER)				
Variances				
HYPER	0.132	0.024	5.577	0.000
"STRETCHINESS FACTOR" - NB FITS BETTER THAN POISSON, WHICH IS INDICATED BY SIG. P-VALUE OR -2ALL. IF POISSON FITS OK DISPERSION SHOULD EQUAL 0)				
Dispersion				
IPV	2.441	0.679	3.594	0.000
Between Level (TERMS FIXED TO 1 OR 0 OMITTED FOR BREVITY)				
FOR EVERY UNIT HIGHER L2 BP HYPER, LOG OF L2 BP IPV IS HIGHER BY .850: MORE L2 MEAN HYPER THAN OTHERS DOES NOT PREDICT MORE L2 MEAN IPV THAN OTHERS				
IPV ON				
FHYPER	0.530	0.786	0.674	0.500
FIXED INTERCEPT FOR HYPER (HYPER IS UNCONDITIONAL SINCE IT IS NOT BEING PREDICTED BY ANYTHING)				
Intercepts				
HYPER	1.436	0.056	25.716	0.000
FIXED INTERCEPT FOR IPV (FOR SOMEONE WITH AVERAGE FHYPER ACROSS DAYS)				
IPV	-3.494	0.653	-5.347	0.000
L2 RANDOM INTERCEPT VARIANCE (NOTHING PREDICTING FHYPER)				
Variances				
FHYPER	0.114	0.024	4.665	0.000
L2 RANDOM INTERCEPT VARIANCE LEFTOVER (IPV IS PREDICTED BY FHYPER)				
Residual Variances				
IPV	1.869	0.666	2.809	0.005

Model 3/4: Negative Binomial/Poisson Hurdle Conditional Distribution

We will use the same latent variable work-around to trick Mplus:

```

TITLE:      Model 3: Negative Binomial Hurdle
                (predict logit of 0, or log of level-1 count if not 0)

DATA:      FILE = Example5.csv;  ! Syntax in same folder as data
VARIABLE:
NAMES = PersonID hyper alc IPV;    ! List ALL variables in data file
USEVARIABLES = hyper IPV;          ! Variables included in this model
MISSING ARE ALL (-99);             ! Missing data identifier
CLUSTER = PersonID;                ! Level-2 ID
COUNT = IPV (nbh);                ! IPV is negative binomial hurdle (if, how much)

ANALYSIS: TYPE = TWOLEVEL RANDOM;  ! In case of random slopes
                ESTIMATOR = MLR;      ! For non-normal hyper
                ALGORITHM = INTEGRATION; ! Because of COUNT option

MODEL:
%WITHIN%
hyper;                ! L1 R: residual variance for hyper
IPV;                  ! L1 dispersion for IPV
IPV ON hyper;         ! L1 WP fixed effect for LOG HOW MUCH IF NOT 0
IPV#1 ON hyper;       ! L1 WP fixed effect for LOGIT of 0 (not 1)

%BETWEEN%
! Move hyper random intercept to new latent variable
Fhyper BY hyper@1;
! Shut off hyper random intercept variance
hyper@0;
! Hyper fixed intercept (hyper=grand-mean-centered)
! IPV: fixed intercepts for amount and for logit of 0
[hyper IPV IPV#1];
! Shut off new latent variable intercept
[Fhyper@0];
! L2 G: random intercept variances
Fhyper IPV IPV#1;
! L2 BP fixed effects using new latent variable
IPV ON Fhyper;      ! L2 BP fixed effect for LOG HOW MUCH IF NOT 0
IPV#1 ON Fhyper;    ! L2 BP fixed effect for LOGIT of 0 (not 1)
IPV WITH IPV#1;     ! L2 G: random intercept covariance

```

The model did not converge, possibly because it could not estimate a IPV dispersion for the “how much” part. So I tricked it into a Poisson-like hurdle model by fixing the dispersion to .001 (now Model 4):

```

%WITHIN%
hyper;                ! L1 R: residual variance for hyper
IPV@.001;             ! Fix L1 dispersion for IPV to mimic Poisson
IPV ON hyper;         ! L1 WP fixed effect for LOG HOW MUCH IF NOT 0
IPV#1 ON hyper;       ! L1 WP fixed effect for LOGIT of 0 (not 1)

```

	Estimate	S.E.	Est./S.E.	Two-Tailed P-Value
Within Level				
GIVEN L1 IPV>0, MORE HYPER THAN USUAL DOES NOT PREDICT HOW MUCH L1 IPV (LOG OF NON-0 COUNT HIGHER BY .054 PER UNIT L1 HYPER)				
IPV ON				
HYPER	0.054	0.131	0.412	0.680
MORE HYPER THAN USUAL --> LESS LIKELY TO BE IPV=0 (SO WILL TRY TO DO IPV) LOGIT OF L1 IPV=0 LOWER BY 1.557 PER UNIT MORE HYPER THAN USUAL				
IPV#1 ON				
HYPER	-1.557	0.443	-3.516	0.000
L1 RESIDUAL VARIANCE (NOTHING PREDICTING HYPER)				
Variances				
HYPER	0.132	0.024	5.576	0.000
Dispersion - FIXED TO MIMIC POISSON HURDLE (NO MORE STRETCHY)				
IPV	0.001	0.000	999.000	999.000
Between Level (TERMS FIXED TO 1 OR 0 OMITTED FOR BREVITY)				
AMONG HITTERS, MORE MEAN HYPER THAN OTHERS DOES NOT PREDICT MORE MEAN IPV THAN OTHERS (LOG OF NON-0 COUNT HIGHER BY .603 PER UNIT L2 HYPER)				
IPV ON				
FHYPER	0.603	0.430	1.402	0.161
MORE MEAN HYPER THAN OTHERS DOES NOT PREDICT TENDENCY TO NOT HIT (LOGIT OF IPV=0 LOWER BY .383 PER UNIT L2 HYPER)				
IPV#1 ON				
FHYPER	-0.383	0.866	-0.442	0.658
THOSE WITH MORE PROBABILITY OF IPV=1 HAVE BIGGER TENDENCY FOR BIG IPV				
IPV WITH				
IPV#1	-0.784	0.303	-2.585	0.010
Intercepts				
HYPER	1.435	0.055	25.881	0.000
FIXED INTERCEPT FOR LOGIT OF IPV=0 WHEN FHYPER=0				
IPV#1	4.159	0.718	5.792	0.000
FIXED INTERCEPT FOR LOG IPV (IF NOT 0) WHEN FHYPER=0				
IPV	0.140	0.251	0.558	0.577
L2 RANDOM INTERCEPT VARIANCE (NOTHING PREDICTING HYPER)				
Variances				
FHYPER	0.115	0.024	4.682	0.000
L2 RANDOM INTERCEPT VARIANCE LEFTOVER (IPV IS PREDICTED BY HYPER)				
Residual Variances				
IPV#1	2.116	0.790	2.677	0.007
IPV	0.291	0.130	2.233	0.026

Model 5: Two-Part Distribution Model (with log transform for continuous part)

```

TITLE:      Model 5: Two-Part
              (predict logit of 1, log of level-1 continuous amount if 1)

DATA:      FILE = Example5.csv;    ! Syntax in same folder as data

DATA TWOPART:    ! Instructs Mplus to cut up IPV into 0/log of amount
NAMES = IPV; BINARY = BIPV; CONTINUOUS = CIPV;
CUTPOINT = 0; TRANSFORM = LOG;    ! Or NONE for no transformation

VARIABLE:
NAMES = PersonID hyper alc IPV;    ! List ALL variables in data file
USEVARIABLES = hyper BIPV CIPV;    ! Variables included in this model
CATEGORICAL = BIPV;                ! Indicate binary outcome
MISSING ARE ALL (-99);              ! Missing data identifier
CLUSTER = PersonID;                ! Level-2 ID

ANALYSIS: TYPE = TWOLEVEL RANDOM;    ! In case of random slopes
              ESTIMATOR = MLR;          ! For non-normal hyper

MODEL:
%WITHIN%
hyper;          ! L1 R: residual variance for hyper
CIPV;           ! L1 R: residual variance for amount IPV
CIPV ON hyper;  ! L1 WP fixed effect for LOG HOW MUCH IF 1
BIPV ON hyper;  ! L1 WP fixed effect for LOGIT of 1 (not 0)

%BETWEEN%
! Move hyper random intercept to new latent variable
Fhyper BY hyper@1;
! Shut off hyper random intercept variance
hyper@0;
! Hyper fixed intercept (hyper=grand-mean-centered)
! IPV: fixed intercept for amount and fixed threshold for logit of 1
[hyper CIPV BIPV$1];
! Shut off new latent variable intercept
[Fhyper@0];
! L2 G: random intercept variances
Fhyper CIPV BIPV;
! L2 BP fixed effects using new latent variable
CIPV ON Fhyper; ! L2 BP fixed effect for LOG HOW MUCH IF 1
BIPV ON Fhyper; ! L2 BP fixed effect for LOGIT of 1 (not 0)
CIPV WITH BIPV; ! L2 G: random intercept covariance

```

	Estimate	S.E.	Est./S.E.	Two-Tailed P-Value
Within Level				
GIVEN L1 IPV>0, MORE HYPER THAN USUAL DOES NOT PREDICT HOW MUCH L1 IPV (LOG OF NON-0 COUNT HIGHER BY .139 PER UNIT L1 HYPER)				
CIPV ON				
HYPER	0.139	0.093	1.494	0.135
MORE HYPER THAN USUAL --> MORE LIKELY TO BE IPV=1 (SO WILL PRY DO IPV) LOGIT OF L1 IPV=1 HIGHER BY 1.589 PER UNIT MORE HYPER THAN USUAL				
BIPV ON				
HYPER	1.589	0.447	3.552	0.000
L1 RESIDUAL VARIANCE (NOTHING PREDICTING HYPER)				
Variances				
HYPER	0.132	0.024	5.577	0.000
L1 RESIDUAL VARIANCE LEFTOVER (IPV IS PREDICTED BY HYPER)				
Residual Variances				
CIPV	0.271	0.039	7.037	0.000
Between Level (TERMS FIXED TO 1 OR 0 OMITTED FOR BREVITY)				
AMONG HITTERS, MORE MEAN HYPER THAN OTHERS DOES NOT PREDICT MORE MEAN IPV THAN OTHERS (LOG OF NON-0 COUNT HIGHER BY .242 PER UNIT L2 HYPER)				
CIPV ON				
FHYPER	0.242	0.271	0.892	0.372
MORE MEAN HYPER THAN OTHERS DOES NOT PREDICT TENDENCY TO NOT HIT (LOGIT OF IPV=1 HIGHER BY .407 PER UNIT L2 HYPER)				
BIPV ON				
FHYPER	0.407	0.878	0.463	0.643
THOSE WITH MORE PROBABILITY OF IPV=1 HAVE BIGGER TENDENCY FOR BIG IPV				
CIPV WITH				
BIPV	0.468	0.212	2.207	0.027
Intercepts				
HYPER	1.436	0.056	25.778	0.000
FIXED INTERCEPT FOR LOG IPV (IF NOT 0) WHEN FHYPER=0				
CIPV	0.458	0.072	6.315	0.000
THRESHOLD*-1 = FIXED INTERCEPT FOR LOGIT OF IPV=1 WHEN FHYPER=0				
Thresholds				
BIPV\$1	4.176	0.730	5.724	0.000
L2 RANDOM INTERCEPT VARIANCE (NOTHING PREDICTING HYPER)				
Variances				
FHYPER	0.114	0.025	4.661	0.000
L2 RANDOM INTERCEPT VARIANCE LEFTOVER (IPV IS PREDICTED BY HYPER)				
Residual Variances				
BIPV	2.092	0.793	2.639	0.008
CIPV	0.120	0.060	2.017	0.044

So which model should we choose to interpret? Unfortunately, relative fit statistics (AIC and BIC) are not comparable across the normal, count-based, and two-part families. What we can do is examine the predicted outcomes for each model and see what seems reasonable. The first plots below (from excel) show the predicted amount of IPV for ± 2 SD of within-person (WP) and between-person (BP) hyperarousal. The WP effect for amount of IPV is significant in the normal and Negative Binomial models. As we can see, the normal model predicts a significant linear relationship, which will eventually extend below 0, whereas the WP effect from other models for the predicted log of the count instead should “shut off” as it approaches 0 (because of the log link transformation—that is its purpose). In contrast, both “if and how much” type models—the Poisson Hurdle and the Two-Part Log—have expected counts that do not approach 0, because that zero-part aspect of the data is modeled as a separate outcome instead. So after dividing the outcome into “0 vs. something”, these two models suggest there is no WP relationship for “something”. In contrast, they have small but significant WP relationships for the “if” part of each model (in the second figures).

